
# NEW FUNCTIONS IN V9.1

CAT concatenates character strings without removing leading or
trailing blanks

CATS concatenates character strings and removes leading and
trailing blanks

CATT concatenates character strings and removes trailing blanks only

CATX concatenates character strings, removes leading and trailing
blanks, and inserts separators

# NEW FUNCTION IN V9.2

CATQ concatenates character or numeric values by using a
delimiter to separate items and by adding quotation
marks to strings that contain the delimiter

the CATX function is similar to the CATQ function except
that CATX does not add quotation marks

included "modifiers" that change the behavior of the
function (no modifiers for other CAT functions)

# results of the CAT, CATS, CATT, and CATX functions are usually
equivalent to results that are produced by certain combinations
of the concatenation operator (||) and the TRIM and LEFT
functions 

Function Equivalent Code (with conCATenation operator ||)

CAT(OF X1‐X4) X1 || X2 || X3 || X4

CATS(OF X1‐X4) TRIM(LEFT(X1)) || TRIM(LEFT(X2)) ||
TRIM(LEFT(X3)) || TRIM(LEFT(X4))

CATS(OF X1‐X4) STRIP(X1) || STRIP(X2) || STRIP(X3) || STRIP(X4)

CATT(OF X1‐X4) TRIM(X1) || TRIM(X2) || TRIM(X3) || TRIM(X4)

CATX(SP, X1‐X4) TRIM(LEFT(X1)) || SP || TRIM(LEFT(X2)) || SP ||
TRIM(LEFT(X3)) || SP || TRIM(LEFT(X4))

CATX(SP, X1‐X4) STRIP(X1) || SP || STRIP(X2) || SP ||
STRIP(X3) || SP || STRIP(X4)

# default length for the CAT, CATS, CATT, and CATX functions is
different from the length that is obtained when you use the
concatenation operator

data names;
input last_name : $25. first_name : $15. @@;
* old way;
full_name_old_way = trim(last_name) || ', ' || first_name;
* new way;
full_name_new_way = catx(', ',last_name,first_name);
datalines;
Zdeb Mike Washington George Squirrel Rocket(J)
;

# NOT JUST FOR CONCATENATION

# ANOTHER USE ... SEARCH FOR VARIABLE VALUES

for example, given a data set
named ANSWERS

create a new data set
containing observations where
at least one answer had a
value of "Y"

data yes;
set answers;
array a(10);
do j=1 to 10 until (a(j) eq 'Y');
end;
if j lt 11;
drop j;
run;

data yes;
set answers;
if find(cat(of a1-a10),'Y');
run;

# CONVENTIONAL
ARRAY AND A
LOOP

# CAT AND FIND
FUNCTIONS

# NEW TASK

create a new data set containing observations where at least
TWO answers had a value of "Y"

data yes;
set answers;
array a(10);
do j=1 to 10 until (found_y eq 2);
  found_y = sum(found_y,(a(j) eq 'Y'));
end;
if found_y eq 2;
drop j found_y;
run;

data yes;
set answers;
if countc(cat(of a1-a10),'Y') ge 2 
run;

# CONVENTIONAL
ARRAY AND A LOOP

# CAT AND COUNTC 
FUNCTIONS

# NEW TASK

add a new variable to each observation in data set ANSWERS 

the variable will contain the number of answers that are "Y"

* array and a loop;
data yes;
set answers;
array a(10);
* use _n_ as loop iterator (no DROP);
do _n_=1 to 10;
   found_y = sum(found_y,(a(_n_) eq 'Y'));
end;
run;

* cat and countc;
data yes;
set answers;
found_y = countc(cat(of a:),'Y'));
run;

# NEW DATA SET WITH NUMERIC ANSWERS

# SIMILAR TASK

create a new data set
containing observations where
at least one answer had a
value of 5

* cat and find;
data five;
set answers;
if find(cat(of a:),'5');
run;

SAS log ...

243  * cat and find;
244  data five;
245  set answers;
246  if find(cat(of a:),'5');
247  run;

NOTE: There were 5 observations
read from the data set
WORK.ANSWERS.
NOTE: The data set WORK.FIVE has 4
observations and 11 variables.

# USE CAT AND FIND

# NO PROBLEM WITH
NUMERIC DATA
(SEARCH FOR '5')

# YES ... YOU COULD HAVE USED ... if max (of a:) eq 5;

# A TEST ...

data five;
set answers;
answers = cat(of a:);
if find(answers,'5');
run;

# WHAT TYPE OF VARIABLE IS ANSWERS ... CHAR OR NUM (WHY?)

# HINT (THE LOG) ...

data x;
input abc @@;
if cat(abc) eq : '111';
datalines;
111222 123456 11111111 987654 111
;

# DIGRESSION ... QUESTION FROM SAS‐L

Is there anyway to check first three digits of numeric value?

# HOW ABOUT THE
CAT FUNCTION

# THIS WORKS

2    data x;
3    input abc @@;
4    if cat(abc) eq : '111';
5    datalines;

NOTE: SAS went to a new line when INPUT statement reached
past the end of a line.
NOTE: The data set WORK.X has 3 observations and 1
variables.
NOTE: DATA statement used (Total process time):
real time           0.59 seconds
cpu time            0.04 seconds

# THE LOG ... NO ERRORS, NO NOTES

# ANOTHER TEST ... THIS DOES NOT WORK (WHY?)

data x;
input abc @@;
if cat(abc) eq : '111';
datalines;
111222 123456 11111111 987654 111 11 1
;

# HINT ... WHAT DOES eq : REALLY MEAN

# A FIX ... if cat(abc,'X') eq : '111';

025 022 
682 401 244 493 
592 401 493
428 493 780 V43 250 
250
414 V45 401 250

# MORE COMPLEXITY ... SEARCH FOR DIAGNOSIS CODES

each observation has up to five,
three character diagnosis codes
and the code 250 indicates
diabetes ... find patient records
with diabetes

data dia_incorrect;
set diag3;
if find(cat(of dx1-dx5),'250') ne 0;
run;

025022
682401244493
592401493
428493780V43250
250
414V45401250

# INCORRECT ... WHY ?

# WHAT IS THE RESULT OF THE CAT
FUNCTION ?

# LOOK AT THE FIRST RECORD ... THERE IS
A 250 BUT NEITHER DIAGNOSIS IS 250
(A COMBINATION OF 025 AND 022)



Searching for Variable Values with CAT Functions

23

data dia_correct;
set diag3;
if find(catx('*',of dx1-dx5),'250') ne 0;
run;

025*022
682*401*244*493
592*401*493
428*493*780*V43*250
250
414*V45*401*250

# CORRECT ... USE CATX

# WHAT IS THE RESULT OF THE CATX
FUNCTION

# LOOK AT THE FIRST RECORD ...
THERE IS NO 250

data dia_ast;
set diag3;
dia = (find(catx('*',of dx1-dx5),'250') ne 0);
ast = (find(catx('*',of dx1-dx5),'493') ne 0);
run;

# CATX MAKES IT EASY TO SEARCH FOR VARIOUS DIAGNOSES

250 indicates diabetes, 493 indicates asthma ... find patient
records with diabetes and/or asthma

# CODE ADDS DUMMY VARIABLES:  1, PRESENT; 0, NOT PRESENT

# DIGRESSION ... QUESTION FROM SAS COMMUNITY

find all observations where all non‐missing
values are 4

create a new variable with a value of 3 for
above, otherwise 9

data test;
set test;
questcat = ifn(verify(catt(of q:),'.4') or sum(of q:) eq . , 9 , 3);
run;

25001 V180
5680  78039 6250  49390 53081
4270  4111  4019  36250 
486   7990  25082 41400 
78659 25000 49320 
49392 4660  2449
34839 2765  40493 4280 

# MORE COMPLEXITY ...
FIVE CHARACTER
DIAGNOSIS CODES,
ONLY THE FIRST THREE
CHARACTERS USED TO
FIND DISORDERS

# ONCE AGAIN ... FIND DIABETES (250) AND/OR ASTHMA (493)

data dia_ast;
set diag5;
array dx(5);
dia = 0;
ast = 0;
do j=1 to 5;
   if dx(j) eq : '250' then dia = 1;
   if dx(j) eq : '493' then ast = 1;
end;
drop j;
run;

# COULD USE AN ARRAY PLUS A LOOP

# EQ :
LOOKS FOR VALUES THAT START WITH THE SPECIFIED STRING



Searching for Variable Values with CAT Functions

29

25001*V180
5680*78039*6250*49390*53081
4270*4111*4019*36250
486*7990*25082*41400
78659*25000*49320
49392*4660*2449
34839*2765*40493*4280

data dia_ast_incorrect;
set diag5;
dia = (find(catx('*',of dx1-dx5),'*250') gt 0);
ast = (find(catx('*',of dx1-dx5),'*493') gt 0);
run;

# OR, USE CATX FUNCTION ... INCORRECT ... WHY ?

# WHAT IS THE RESULT OF THE
CATX FUNCTION ?

# LOOK AT THE FIRST RECORD ...
STARTS WITH 250

data dia_ast_correct;
set diag5;
dia = (find(catx('*','*',of dx1-dx5),'*250') gt 0);
ast = (find(catx('*','*',of dx1-dx5),'*493') gt 0);
run;

**25001*V180
**5680*78039*6250*49390*53081
**4270*4111*4019*36250
**486*7990*25082*41400
**78659*25000*49320
**49392*4660*2449
**34839*2765*40493*4280

# MODIFIED CATX FUNCTION

# WHAT IS THE RESULT OF THE
CATX FUNCTION

# LOOK AT THE FIRST RECORD ...
STARTS WITH  **250

# "REAL LIFE" EXAMPLE

data set with 2.5 million observations ... each observation has an
admit, principal, and up to fourteen other diagnoses

identify observations with either:  diabetes (250); asthma (493);
heart attack (410); female breast cancer (174); influenza (487)

* use an array and a loop;
data array_loop;
set temp;
array dx(16) adx pdx odx1-odx14;
dia = 0;
ast = 0;
ami = 0;
brc = 0;
flu = 0;
do j=1 to 16 until (dx(j) eq ' ');
   if dx(j) eq : '250' then dia = 1;
   if dx(j) eq : '493' then ast = 1;
   if dx(j) eq : '410' then ami = 1;
   if dx(j) eq : '174' then brc = 1;
   if dx(j) eq : '487' then flu = 1;
end;
drop j;

# DONE THREE
DIFFERENT
WAYS

* CAT and FIND with a new variable
 (and a LENGTH statement);

data cat_find_new;
set temp;
length string $100;
string = catx('*','*',adx,pdx,of odx:);
dia = (find(string,'*250') gt 0);
ast = (find(string,'*493') gt 0);
ami = (find(string,'*410') gt 0);
brc = (find(string,'*174') gt 0);
flu = (find(string,'*487') gt 0);
drop string;
run;

* CAT and FIND with repeated concatenation;

data cat_find_repeat;
set temp;
dia = (find(catx('*','*',adx, pdx,of odx:),'*250') gt 0);
ast = (find(catx('*','*',adx, pdx,of odx:),'*493') gt 0);
ami = (find(catx('*','*',adx, pdx,of odx:),'*410') gt 0);
brc = (find(catx('*','*',adx, pdx,of odx:),'*174') gt 0);
flu = (find(catx('*','*',adx, pdx,of odx:),'*487') gt 0);
run;



Searching for Variable Values with CAT Functions

35

# EACH METHOD TRIED 25 TIMES

# DATA SET (NAMED TEMP) IN MEMORY USING SASFILE
sasfile temp load;

# TIMES ARE IN  SECONDS

METHOD
ELAPSED
TIME

CPU
 TIME

ARRAY AND LOOP 21.1 17.4

CAT AND FIND WITH A NEW VARIABLE 19.9 16.4

CAT AND FIND WITH A NEW VARIABLE (AND A LENGTH
STATEMENT) 18.4 13.3

CAT AND FIND WITH REPEATED CONCATENATION 18.6 14.3

# WHAT IF SELECTION RULES ARE COMPLEX

Traumatic brain injury includes ICD‐9 codes 800‐801.9, 803‐804.9,
850‐854.1, 950.1‐950.3, 959.01 and 995.55.

proc format;
value $tbi
'800'‐'80199', '803'‐'80499' , '850'‐'85419', 
'9501'‐'95039', '95901' , '99555' = '1'
other = '0';
run;

# CONCATENATE FORMATTED VALUES OF VARIABLES

filename nosee dummy;

data tbi;
set temp;
file nosee;
put (adx pdx odx1‐odx14) ($tbi.) @;
if find(_file_,'1') then output;
put;
run;

# SELECTION RULES NOT COMPLEX ... ANOTHER APPROACH FOR
LOOKING AT THE FIRST THREE CHARACTERS IN FIVE CHARACTER
DX CODES ... CONCATENATE FIRST THREE CHARACTERS

filename nosee dummy;

data tbi;
set temp;
file nosee;
put (adx pdx odx1‐odx14) ($3. +1) @;
if find(_file_,'250') then output;
put;
run;

# LOOK FOR MULTIPLE DIAGNOSES ... SAVE PATTERN

proc format;
value $dia_ast
'250' ‐ '25099' = 'D' '493' ‐ '49399' = 'A'  other = 'x';
run;

data dia_ast;
length pattern $16;
set temp;
file nosee;
put (adx pdx odx1‐odx14) ($dia_ast.) @;
pattern = _file_;
if findc(_file_,'DA') then output;
put;
run;

# COMPARE

# LAST DIGRESSION ... FROM SAS COMMUNITY

if I concatenate several data sets, is there any way to find out what
data set an observation came from

# VERSION 9.2+ ... CAN USE INDSNAME

data all;
set one two three indsname=temp;
dsn = temp;
run;

# VERSION 9.13

data all;
set one (in=_1) two (in=_2) three(in=_3);
from = find(catt(of  _:),'1');
run;

CONTACT INFORMATION ...

Mike Zdeb
msz03@albany.edu
518 402 6479


